
Scalable formal verification constitutes an important challenge for the design of asynchronous cir-
cuits. Deadlock freedom is a property that is desired but hard to verify. It is an emergent property
that has to be verified monolithically. We present our approach to using ACL2 to verify necessary
and sufficient conditions over asynchronous delay-insensitive primitives. These conditions are used
to derive SAT/SMT instances from circuits built out of theseprimitives. These SAT/SMT instances
help in establishing absence of deadlocks. Our verificationeffort consists of building an executable
checker in the ACL2 logic tailored for our purpose. We prove that this checker is correct. This ap-
proach enables us to prove ACL2 theorems involvingdefun-sk constructs and free variables fully
automatically.

1 Introduction

Today’s hardware designs commonly are clocked. A rhythmic clock signal ensures that a designer can
assume a discrete notion of time. The clocked design paradigm has many advantages, but they come at
a high cost. It induces overhead and delay in terms of speed, data flow and energy [6]. In a clock-free or
asynchronousdesign each element acts only when necessary and at its own pace. This can save energy,
can increase speed and can decrease latency of communications.

Recently, Click has been proposed as a library for the designof asynchronous circuits [4]. It consists
of primitives that aredelay-insensitive, i.e., primitives that behave correctly regardless of any delay
induced by interfacing with the environment. Click primitives are low-level hardware design templates
for delay-insensitive elements such as storages, forks, joins and distributors1. Connected in a pipelined
fashion, the purpose of these primitives is to behave as “lego-like” as possible. They restore a high level
of abstraction during the design phase, even when a close link to realistic asynchronous hardware is
maintained.

Many state-of-the-art formal verification efforts on asynchronous circuits focus on proving properties
over elements in isolation [11, 3, 10]. Deadlock freedom, however, is an emergent property. Establishing
deadlock freedom of primitives in isolation does not provide any information on deadlock freedom of the
entire system. A monolithic approach is mandatory. Our approach is to automatically derive SAT/SMT
instances from Click circuits. If the instance is infeasible, the circuit is deadlock-free. If a solution
is found, this solution corresponds to a structural deadlock. This approach has been applied before to
synchronous circuits, where it shows great promise in termsof scalability [7, 1].

Consider the network in Figure 1 as an example. The circuit iscomposed of six Click primitives.
These primitives use handshakesa through f to establish mutual communication. The input injects
packets which are duplicated by the fork. Two storagess0 ands1 buffer these packets. The join waits for
two packets at its inputs and combines them into one packet, which is sent to the output.

1To be more precise, the primitives are quasi-delay-insensitive. For sake of presentation, we do not distinct these terms.

Figure 1: Click circuit

Given this circuit, we automatically derive the following result:

Dead(a) ⇐⇒ ((s0∧¬s1)∨ (s1∧¬s0))∧ (s0 = s1) (1)

In words, this formula states that there is a deadlock involving handshakea if and only if exactly one
of the storages is fulland the internal state of both storages is equal. The left hand side of the conjunct
indicates that if, e.g., storages0 contains a packet but storages1 does not, a deadlock would occur. In
this configuration, the fork will never be able to duplicate two packets, whereas the join will never be
able to combine two packets. The right hand side indicates that invariably both storages will either both
be empty or both be full. The formula is not satisfiable, i.e.,there is no assignment of values to variables
that makes the formula true. Consequently, there is no deadlock.

Key to deriving a deadlock formula such as Equation 1, is to establish necessary and sufficient con-
ditions for each Click primitive in isolation. These conditions must characterize the reasons that cause a
handshake to beblocked, i.e., not able to transmit a packet, oridle, i.e., not receiving a packet. We use
the join as a running example. The join waits for data from itstwo inputsa andb before forwarding data
to outputc. Input a is permanently blocked if and only if one of two cases arise. First, when outputc
is permanently blocked the join can never forward a packet. It therefore blocks inputa. Secondly, when
no packet arrives at inputb, the join will never be able to merge two packets and will never produce an
output. Inputa is blocked. Hence the join induces the following necessary and sufficient condition:

Complete Condition. The input of a join is permanently blocked if and only if either its output is per-
manently blocked or the other input is permanently idle.

Block(a) ⇐⇒ Block(c)∨ Idle(b)

Correct necessary and sufficient conditions for each primitive in the Click library are vital to the
correctness of our approach. Even though their correctnessis often seems obvious, their formalizations
are complicated and their proofs of correctness are often highly tedious. Moreover, the Click library
contains many primitives and our approach requires multiple necessary and sufficient conditions per
primitive. Therefore, we have implemented a small and highly tailored checker for Click primitives in
ACL2. This checker is able to automatically verify necessary and sufficient conditions built out of block-
and idle predicates for a library of delay insensitive primitives. This paper presents ACL2 details of our
verification effort, a broader overview can be found in our publication at ASYNC [8]. Details on how
these conditions can be used to build a formula such as Equation 1 can be found elsewhere [7, 1].

2 Formalizing Blocking and Idle Conditions

We represent Click primitives using the eXtended Delay Insensitive (XDI) specification [9, 2]. In this
paper, XDI specifications are represented using automata. We first introduce the parts of the XDI formal-
ism relevant to this paper. Then, the execution semantics ofXDI state machines are formalized. Finally,

we use Linear Temporal Logic (LTL) to formalize properties over executions of Click circuits. LTL uses
the G(lobally) operator to express that some property is always true, and theF(inally) operator which
expresses that some property is eventually true [5].

2.1 Formalization of Click Primitives

A Click primitive is connected to several other primitives (its environment) and may use several hand-
shakes for this. Each handshake is implemented by two wireshR andhA for requests and acknowledg-
ments. Each wire is either an input to the primitive, or an output. We allow the possibility that a request
for handshakeh is accompanied by datad. In this case, the handshake will be denoted withhd.

Running Example, Part 1. For the join, the set of handshakes is{a,b,c}. The set of input wires is
{aR,bR,cA} and the set of output wires is{aA,bA,cR}. The possibility of transmitting data with requests
is not needed.

Figure 2: XDI state graph of a join

Figure 2 shows the XDI state machine of the join. The following ACL2 code, which will be explained
in more detail hereafter, corresponds to this XDI state graph:

(defconst *xdi-sm-join*

’(;;State Init Type Transitions

(s0 T BOX (((b R I) s2) ((a R I) s1))))

(s1 NIL BOX (((b R I) s3)))

(s2 NIL BOX (((a R I) s3)))

(s3 NIL TRANSIENT (((c R O) s4)))

(s4 NIL BOX (((c A I) s5)))

(s5 NIL TRANSIENT (((b A O) s7) ((a A O) s6)))

(s6 NIL TRANSIENT (((b A O) s0) ((a R I) s9)))

(s7 NIL TRANSIENT (((a A O) s0) ((b R I) s8)))

(s8 NIL TRANSIENT (((a A O) s2)))

(s9 NIL TRANSIENT (((b A O) s1))))

An XDI specification consists of a set of states. There is exactly one state that is the initial state.
In contrast to the full XDI specification, which provides fivedifferent types of states, our presentation

allows only two types of states:indifferentstates (denoted withBOX) andtransientstates. An indifferent
state poses no progress obligation on either the circuit or its environment. A transient state requires
progress of the circuit, i.e., the primitive eventually hasto proceed to a next state.

Running Example, Part 2. For the join, state s0 is the initial state. As this state requires an input
from the environment on wires aR and bR, there are no progress obligations and the type of state s0 is
indifferent. In state s3, requests have been received from both a and b. The circuit has to send a request
to its output c. Consequently, the type of state s3 is transient.

A transition(w s) is a tuple containing the labelw that represents the wire on which a communi-
cation is to occur for the transition to the next states to happen. A wirew is represented by a tuple
(h R/A I/O) with three values representing the handshake, whether the wire is used for Requests or
Acknowledgments, and whether the wire is an input or and output to the primitive. For example, if the
join is in its initial state and the input wireaR changes from low to high, the join moves to states1. For
details on rules on which transitions are allowed and required in XDI specifications, we refer to papers
on the XDI formalism (e.g., [9]).

2.2 Execution Semantics

The execution semantics of an XDI state machineX are formalized relative to its environment. Since the
environment consists of Click primitives, it is basically alarge XDI state machine. The only information
relevant to the analysis of primitiveX, is whether its input wires arestableor not. A wire w is stable
if and only if its value is permanently unchanged. This implies that if wirew is stable, no transition
labelled withw occurs. Therefore, the environment, i.e., the complete setof Click primitives constituting
the circuit, is represented as a set of input wires such that each wire in the set is deemed to be stable.

Running Example, Part 3. We consider an environment of the join in which wire cA is stable. Any
execution will strand in state s4, i.e., waiting for the environment to acknowledge the receipt of data
by output c after a request to c has been sent to fetch this data. Essentially, the join is dead because
the environment permanentlyblockson handshake c. In another environment wires aR and bR may be
stable. The join will get stuck in its initial state s0. It is waiting for the environment to send requests.
Essentially, the join is dead because the environment is permanentlyidle on handshakes a and b. In total,
the three input wires induce23 different possible groups of environments of interest while analyzing the
join, ranging from a live one (i.e., the environment is the empty set), to an environment where all three
input wires of the join are stable (i.e., the environment is the set{aR,bR,cA}).

First, we define a predicate to indicate that a given wire is aninput wire. We make use of theproj
function, which returns thenth projection in a list of lists. For example,(proj 0 *xdi-sm-join*)

return the set of states of the XDI state machine of the join.

(defun input-wirep (xdi-sm wire)

(member-equal (list (car wire) (cadr wire) ’I)

(proj 0 (union (proj 3 xdi-sm)))))

Functioninput-wirep takes as input an XDI state machine and a partial descriptionof a wire, namely
a tuple(h R/A), whereh is the handshake andR/A indicates a Request or Acknowledge. This is trans-
formed to a wire(h R/A I), for which is searched in the set of labels on the transitionsof the XDI state
machine. An environment can now be defined as a set of input wires.

(defun envp (xdi-sm env)

(if (endp env)

t

(and (input-wirep xdi-sm (car env))

(envp xdi-sm (cdr env)))))

The actual value of the environment depends on the state of the network. In the remainder of this paper,
we quantify overall possible environments.

Given an environment, we can define the next step function of XDI state machines. Given a current
state, this function returns the set of next possible states. First, we define a function that takes as input
a list of transitionsts and filters out transitions labelled with a stable wire, i.e., transitions that cannot
occur. Functionstable returnst if and only if the given wire is stable in the given environment.

(defun remove-stable-wire-transitions (ts env)

(cond ((endp ts)

nil)

((stable (caar ts) env)

(remove-stable-wire-transitions (cdr ts) env))

(t

(cons (car ts)

(remove-stable-wire-transitions (cdr ts) env)))))

This function yields all transitions((h R/A I/O) s’) that are not stable. The next step function gives
it all possible transitions from the current states and takes from each resulting transition the next states’.

(defun xdi-step (xdi-sm s env)

(proj 1 (remove-stable-wire-transitions (nth 3 (assoc s xdi-sm)) env)))

2.3 Labelling States as Blocking or Idling

We identify each non-transient state asblockingor idling with respect to handshakeh. We define these
labels in such a way that if primitiveX is permanently stuck in a state labelled as “blockingh”, handshake
h is permanently blocked. Handshakeh is permanently idle, if primitiveX permanently remains in a state
labelled “idlingh”.

Running Example, Part 4. For the join, we consider handshake a, which uses wires aR and aA to
communicate with the join. States s1, s3, s4, s5, s7, s8 and s9 are blocking this handshake. In these states,
handshake a has sent a request to the join, which has not been acknowledged by the join yet. If the join
is permanently stuck in these states, handshake a will permanently wait for an acknowledgment from the
join. Handshake a is permanently blocked. The remaining states s0, s2, and s6 are idling handshake a.
In these states, the join waits for a request from handshake a. When it is permanently stuck in of these
states, handshake a is failing to send this request. Handshake a is permanently idle.

To define predicatesblocking andidling, we define an executable functioncompute-b/i which
recursively explores the state machine and returns an association list mapping to each state a Boolean
value indicating how the state should be labelled. The intuition of this function is that initially states
are idling. As soon as a transition((h R I/O) s’) occurs, apparently the primitive is in a state where
it has been requested to communicate on handshakeh, but has not finished this communication yet.
All subsequent states are therefore blocking handshakeh, until a transition((h A I/O) s’) occurs.
After this transition, the primitive has successfully dealt with the request and no communication occurs
on handshakeh. All subsequent states are idling handshakeh. This repeats, until all states have been
explored.

(mutual-recursion

(defun compute-b/i (xdi-sm s h flg ret)

(if (assoc s ret)

ret

(let ((ret (acons s flg ret)))

(compute-b/i-ts xdi-sm (nth 3 (assoc s xdi-sm)) h flg ret))))

(defun compute-b/i-ts (xdi-sm ts h flg ret)

(let ((ret (cond ((equal (caaar ts) h)

(compute-b/i xdi-sm (cadar ts) h (not flg) ret))

(t

(compute-b/i xdi-sm (cadar ts) h flg ret)))))

(compute-b/i-ts xdi-sm (cdr ts) h flg ret))))

Figure 3: Implementation ofcompute-b/i

Figure 3 shows the ACL2 code of functioncompute-b/iwhich computes this association list. Func-
tion compute-b/i takes as second parameter a states. The third parameter is a flag indicating whether
currently explored states are to be marked blocking or idling. It checks whether this state has already
been explored. If so, then no further exploration is needed.Otherwise, it updates the returned association
list ret by associating the current value of the flag to the current state. After this update, the function
recursively explores all transitions leading out of the current state. Functioncompute-b/i-ts takes as
second parameter a set of transitions. Sequentially two things occur. First, the first transition of the set
is analyzed. If this transition concerns handshakeh, the flag is changed indicating that a switch from
blocking to idling happens, or the other way around. A recursive call with the next state as value fors is
performed. Second, the remaining transitions are recursively explored.

Functioncompute-b/i is initially called with the initial state and as flag the value nil. The pred-
icate blocking can now be defined by simply looking up the given state in the result of function
compute-b/i.

(defun blocking (xdi-sm s h)

(cadr (assoc s (compute-b/i xdi-sm (xdi-get-init-state xdi-sm) h nil nil))))

Predicateidling is defined as not blocking.

Running Example, Part 5. The state machine of the join contains a transition s0
aR−→ s1. We can compute

that:
(blocking *xdi-sm-join* s0 a)

evaluates tonil, whereas
(blocking *xdi-sm-join* s1 a)

evaluates tot.
This represents that state s0 is idling handshake a, whereas state s1 is blocking handshake a.

Remarks

An important assumption on the Click primitives is that their XDI specification ensures that function
blocking is uniquely defined over all non-transient states. If a certain states can be reached from the
initial state using a sequence of transitions with one transition labelledhR and no transition labelledhA,
functionblocking enforces blocking(h,s) to be true. If this statescan also be reached with a sequence

of transitions withouthR as label, functionblocking enforces blocking(h,s) to be false. Such state
graphs are not allowed. We will call a Click primitive for which function blocking is unique over all
non-transient statesunambiguous. In ACL2, we have an executable function checking for unambiguity.

Functioncompute-b/i does not necessarily terminate. To prove termination, we require both a list
of assumptions and some checks which have to be performed by the function before each recursive call.
We have added the assumption as guards, and defined a logical version of this function with the additional
checks. For the logical version, we have proven termination. The code shown here is the executable
version, without these checks. Using anmbe-construct, we have proven that under assumption of the
guards, the logical and the executable versions are equivalent.

2.4 Formulating Block- or Idle Conditions

Whether a primitive can be stuck in a blocking- or idling state depends on the environment. Consider
again the state machine of the join (see Figure 2). If the environment dictates that wirecA is stable,
any execution will strand in states4. This state is blocking handshakea and therefore handshakea is
permanently blocked. We say that a handshakeh is permanently blocked if and only if a primitive will
eventually get stuck in non-transient states labelled “blocking h”.

To define LTL properties over XDI state machines, we first define the notion of trace. A trace is a set
of states that is connected via thexdi-step function.

(defun xdi-tracep (xdi-sm trace env)

(cond ((endp trace)

t)

((endp (cdr trace))

t)

(t

(and (member (cadr trace) (xdi-step xdi-sm (car trace) env))

(xdi-tracep xdi-sm (cdr trace) env)))))

To express that a machine is permanently stuck in blocking states, we use adefun-sk construct to
quantify over all possible traces starting in the current state.

(defun-sk G-blocking_ (xdi-sm h s env)

(forall (trace)

(implies (and (xdi-tracep xdi-sm trace env)

(equal (car trace) s))

(or (equal (nth 2 (assoc (car (last trace)) xdi-sm))

’transient)

(blocking xdi-sm (car (last trace)) h)))))

The trailing underscore is used to indicate that the function is non-executable. Any trace starting ins
ends either in a transient state or in a state that is blockinghandshakeh. Note that we deal with finite
traces only. Since the XDI automata are always finite, any infinite trace consists of a prefix followed
by a repetition of some trace induced by a cycle. It it therefore sufficient to analyze all finite – but of
unbounded length – prefixes.

Similarly, we express theF operator using adefun-sk construct introducing an existential quantifier.

(defun-sk F-G-blocking_ (xdi-sm h s env)

(exists (trace)

(and (xdi-tracep xdi-sm trace env)

(equal (car trace) s)

(G-blocking_ xdi-sm h (car (last trace)) env))))

Similar definitions have been formulated for idling. Given environmentenv, handshakeh is permanently
blockedif and only if the corresponding XDI state machine is eventually always in a blocking state. Sim-
ilarly, handshakeh is permanently idleif and only if the corresponding XDI state machine is eventually
always in an idling state.

(defun Blocked_ (xdi-sm h env)

(F-G-blocking_ X h (xdi-get-init-state xdi-sm) env))

(defun Idle_ (xdi-sm h env)

(F-G-idling_ xdi-sm h (xdi-get-init-state xdi-sm) env))

Finally, we can formulate necessary and sufficient conditions per Click primitive. For example, the
ACL2 formalization of the running example becomes:

Theorem 1.

(defthm blocking-equation-join

(implies (envp xdi-sm env)

(iff (Blocked_ *xdi-sm-join* ’a env)

(or (Blocked_ *xdi-sm-join* ’c env)

(Idle_ *xdi-sm-join* ’b env)))))

3 Model Checking Blocking and Idle Conditions

Proving Theorem 1 could be done manually as follows. First, acase distinction is required over all
possible environments (in this case, eight in total). For each environment, all traces have to be explored
to see whether the labels computed by functioncompute-b/i always satisfy the formula that is to be
proven. As we need multiple theorems per primitive and thereis a whole library of Click primitives, we
want to prove dozens of theorems such as Theorem 1. Naturally, a manual proof is simply infeasible
and an automated approach is mandatory. Therefore, our proof technique is to A.) define executable
counterparts to non-executable functionsBlocked andIdle and B.) make an automatic enumeration
of all possible environments. A once and for all proof that these functions are correctly implemented can
then be used to prove Theorem 1 without further interaction.

3.1 ACL2 Overview of Automated Proof

Our objective for Part A. is to implement functionBlocked in such a way that the following lemma can
be proven (similar forIdle). This is the specification of functionblocked, its definition will follow.

Lemma 1.

(defthm rewrite-non-exec-Blocked_-to-exec-Blocked

(implies (and (xdi-smp xdi-sm)

(member s (proj 0 xdi-sm)))

(equal (Blocked_ xdi-sm h s env)

(Blocked xdi-sm h s env))))

:rule-classes :definition)

Functionxdi-smp recognizes syntactically valid XDI state machines. For anystates that is a valid
state, the result of functionBlocked is equivalent to that of its specificationBlocked . Once this theory
has been established, the non-executable definitionBlocked is disabled in the theory.

(in-theory (disable Blocked_))

This way we are sure – when proving a theorem – that any occurrence ofBlocked will be rewritten
using Lemma 1only (note that a defun-sk construct is non-executable, but is still often rewritten to a goal
without the original function name, thereby preventing application of Lemma 1).

As for Part B., we straightforwardly implement a functionreasonable-envs which takes as input
an XDI state machine and generates a list of all possible environments. We first implement function
compute-input-wires which given an XDI state machine returns the set of input wires. The set of
environments is then computed as follows:

(defun reasonable-envs (xdi-sm)

(powerset (remove-duplicates (compute-input-wires xdi-sm))))

The set of input wires is assembled. Duplicate entries are removed, for sake of efficiency. The list of
relevant environments contains any subset of these wires.

Remark

The number of environments grows exponentially. The XDI automata of interest are however not very
large. In Section 4 we apply our method to a non-trivial Clickprimitive, namely the distributor. Regard-
less of the large number of environments, we can easily deal with this primitive.

Using Parts A. and B., we can prove Theorem 1 completely automatically, after rewriting it slightly.
Our final formulation becomes:

(defthm blocking-equation-join

(and (xdi-smp-guard *xdi-sm-join*)

(implies (member env (reasonable-envs *xdi-sm-join*))

(iff (Blocked_ *xdi-sm-join* nil ’in0 env)

(or (Blocked_ *xdi-sm-join* nil ’out env)

(Idle_ *xdi-sm-join* nil ’in1 env))))))

First, we explicitly verify that the constant*xdi-sm-join* satisfies all guards necessary for cor-
rect execution of functionsBlocked andIdle. Functionxdi-smp-guard is executable and therefore
(xdi-smp-guard *xdi-sm-join*) is proven without further interaction. Secondly, we reformulate
the theorem, so that open variableenv is a member of a computable set of environments. The ACL2
simplifier will compute all reasonable environments. Subsequently, having the following lemma enabled
in the theory ensures that themember construct breaks the goal down into eight different subgoals (one
for each environment):

(defthm member-rewrite

(equal (member a (cons b x))

(if (equal a b) (cons b x)

(member a b))))

For each subgoal, the ACL2 simplifier uses Lemma 1 (and a similar lemma forIdle) to rewrite the
subgoal to executable versions ofBlocked andIdle. At this point, all functions are executable and
there are no variables. The truth of each subgoal is automatically evaluated.

3.2 Implementation ofBlocked and Idle

The implementation ofBlocked needs to check whether eventually generally the machine is in states
that are either transient or labelled as “blocking” by function compute-b/i. So first a function must be
implemented which decides whether in a certain start states the machine generally is in such states, i.e.,
a functionG-blocking must be implemented. Figure 4 shows the implementation of this function.

(mutual-recursion

(defun G-blocking (xdi-sm visited s h env)

(if (not (member-equal s visited))

(if (or (equal (nth 2 (assoc s xdi-sm)) ’transient)

(blocking xdi-sm s h))

(G-blocking-ss xdi-sm (cons s visited)

(remove-equal s (xdi-step xdi-sm s env)) h env)

nil)

visited))

(defun G-blocking-ss (xdi-sm visited ss h env)

(if (endp ss)

visited

(let ((ret (G-blocking xdi-sm visited (car ss) h env)))

(cond ((equal ret nil)

ret)

(t

(let ((ret2 (G-blocking-ss xdi-sm visited (cdr ss) h env)))

(if ret2

(append ret ret2)

nil))))))))

Figure 4: Implementation ofG-blocking

FunctionG-blocking takes as input the XDI state machine, an accumulator of visited states, the
current state, the handshake and the current environment. If it returnsnil this indicates that there is
some reachable state that is not labelled “blocking” with respect to handshakeh. If it does not returnnil
it does not returnt, but instead it returns a list of states that has been explored. This will be used in the
proofs later on.

If the current states is not visited and labelled “blocking” with respect to handshakeh, exploration
continues with all next states using functionG-blocking-ss. This function deals with two cases: either
there are no states to be explored, or there are states to be explored. In the first case, as there are no
more reachable states, all reachable states have been explored. Therefore, the function should returnt,
but instead returns the accumulatorvisited storing all explored states. Otherwise, the function checks
whether the first state in the list isG-blocking. If the result of this check isnil, this result is returned.
Otherwise, the intermediate result (i.e., the states explored in the recursive call) are appended to the final
result.

A similar function is implemented to computeF-G-blocking . We extend both functions with an
extra flag so that these functions can also be used to compute(F-)G-idling. Using these functions, we

can define the implementations ofBlocked andIdle .

(defun Blocked (xdi-sm h env)

(consp (F-G-blocking xdi-sm nil ‘(,(xdi-get-initial-state xdi-sm)) h env)))

(defun Idle (xdi-sm h env)

(consp (F-G-idling xdi-sm nil ‘(,(xdi-get-initial-state xdi-sm)) h env)))

3.3 Proof of Lemma 1

We present the proof of correctness of functionG-blocking. The proofs forF-G-blocking are similar.
The proof is in two directions. We prove Lemma 2 which states that for any state such that specification
G-blocking returnst, executable functionG-blocking returns a non-empty list of visited states. Sec-
ondly, we prove Lemma 3 which states that any state for which executable functionG-blocking returns
a non-empty list, specificationG-blocking returnst.

Lemma 2.
We formulate the lemma in such a way that induction overG-blocking is possible. This requires

parametersvisited andss to be free. Also, any assumption on these variables must be aninvariant
over functionG-blocking. Our formalization is as follows, and will be detailed hereafter:

(defthm G-blocking_-->G-blocking

(implies (and (xdi-smp xdi-sm)

(G-blocking_ xdi-sm h s env)

(consp ss)

(subsetp ss (proj 0 xdi-sm))

(A-xdi-reachable xdi-sm env s ss))

(G-blocking xdi-sm visited ss h t env))

Assume a valid XDI state machine and a states which is generally blocking according to the specifica-
tion. For any non-empty setss of valid states we prove that executable functionG-blocking returns
a non-empty list. As an invariant, we require that all statesin ss are reachable from states. Function
A-xdi-reachable returnst if and only if all states inss are reachable from states. Reachability be-
tween two statess1 ands2 is straightforwardly defined using adefun-sk construct as the existence of
a non-empty trace starting ins1 and ending ins2:

(defun-sk xdi-reachable (xdi-sm env s1 s2)

(exists (trace)

(and (xdi-tracep xdi-sm trace env)

(consp trace)

(equal (car trace) s1)

(equal (car (last trace)) s2))))

Once it is proven that reachability of statesss from states is indeed an invariant, the proof becomes
conceptually very easy. As soon as G-blocking encounters a non-transient states2, we know from the
invariant that states2 is reachable from states. From assumption(G-blocking xdi-sm h s env)

we can prove thats2 is blocking, which is expressed by lemmaspec-of-G-blocking :

(defthm spec-of-G-blocking_

(implies (and (G-blocking_ xdi-sm h s env)

(xdi-reachable xdi-sm env s s2)

(not (equal (nth 2 (assoc s2 xdi-sm)) ’transient)))

(blocking xdi-sm s2 h))

:hints (("Goal" :use ((:instance G-blocking_-necc

(trace (xdi-reachable-witness xdi-sm env s s2)))))))

The theorem is proven by instantiating the theorem introduced by thedefun-sk event corresponding to
G-blocking . This instantiation requires a trace froms to s2. This trace is exactly the witness created
by thedefun-sk event corresponding toxdi-reachable.

Now we do induction overG-blocking. We have to prove that it does not returnnil. It returnsnil
only if it encounters an illegal state or a non-transient state that is not labelled “blocking”. The first case
cannot happen due to assumption(xdi-smp xdi-sm). The second case cannot happen since the invari-
ant ensures that any explored state is reachable from states and since lemmaspec-of-G-blocking
can be used to prove that any state reachable froms is labelled “blocking”.

Lemma 3.
The lemma is formulated as follows:

(defthm G-blocking-->G-blocking_

(implies (and (xdi-smp xdi-sm (cars xdi-sm))

(member-equal s (cars xdi-sm))

(G-blocking xdi-sm nil (list s) h env))

(G-blocking_ xdi-sm h s env)))

Given a valid XDI state machine and a valid states, the executable function correctly decides the LTL
formula for states.

For the proof of the lemma we use the fact that functionG-blocking does not returnt when all
reachable non-transient states are labelled “blocking”, but the accumulatorvisited instead. As we
reason over reachable states, we define functionxdi-reach which takes as input a set of statesss and
assembles all states reachable from any state inss. The reasoning is as follows:

1. Any state returned byG blocking is either transient or labelled “blocking”.

2. For any set of statesss, the set of states accumulated by
(G-blocking xdi-sm visited ss h env) contains all states returned by(xdi-reach ss).

3. Any state reachable from any state inss is in (xdi-reach ss).

4. We have to prove that any states2 reachable froms is either transient or labelled “blocking”.
By 3.) it follows that:
(member s2 (xdi-reach (list s))

By 2.) it follows that:
(member s2 (G-blocking xdi-sm nil (list s) h env)

By 1.) it follows thats2 is either transient or labelled “blocking”.

First, we prove a theorem stating that any non-transient state returned byG-blocking is indeed
labelled “blocking”.

(defthm all-states-in-G-blocking-are-blocking

(implies (A-blocking xdi-sm visited h)

(A-blocking xdi-sm (G-blocking xdi-sm visited ss h env) h)))

FunctionA-blocking is a universal quantifier expressing that all given states are either transient or
labelled “blocking”. Assuming this property holds for all initially accumulated states, this property
holds for accumulated states.

We then prove thatG-blocking returns all states assembled byxdi-reach.

(defthm all-reachable-states-in-G-blocking

(implies (G-blocking xdi-sm visited ss h env)

(subsetp (xdi-reach xdi-sm visited ss env)

(G-blocking xdi-sm visited ss h env))))

Assuming thatG-blocking does not returnnil, we prove that any state that is in the reach of some state
in ss is also member of the list of accumulated states returned byG-blocking.

Finally, we prove correctness of functionxdi-reach, i.e., that it contain all reachable states.

(defthm spec-of-xdi-reach

(implies (and (xdi-smp xdi-sm (proj 0 xdi-sm))

(member s (proj 0 xdi-sm))

(xdi-reachable xdi-sm env s s2))

(member-equal s2 (xdi-reach xdi-sm nil (list s) env)))

Using these lemmas, Lemma 3 can be proven without induction.To prove the universal quantifier
introduced byG-blocking , we have to prove of a witness state

sw = (G-blocking -witness xdi-sm h s env)

that it is either transient or labelled “blocking”. Instantiating the first lemma with the accumulator
visited set tonil, automatically discharges its assumption. What remains tobe proven is thatsw

is accumulated byG-blocking. This is proven by the second lemma, instantiated withvisited set to
nil andss set to(list s). This forces us to prove that statesw is a member ofxdi-reach. The third
lemma is used to prove this, instantiatings2 with (G-blocking -witness xdi-sm h s env).

4 Application

The distributor (see Figure 5) is a Click primitive used for routing packets through a network. It uses
handshakea on which the availability of data is communicated and three handshakes selectd (00≤ d ≤
11). If d = 00, the incoming data is dropped. Ifd = 01, the packet is routed towards outputb. Similarly
d = 10 routes towards outputc. Figure 6 shows the XDI state graph of the distributor. The set of

a

select

b

c

Figure 5: Schematic overview of the distributor

input wiresWI is {aR,selectd
R
,bA,cA}. The remaining wires are output, i.e.,WO = {aA,selectA,bR,cR}.

Handshake select uses data for requests.
The blocking equation of inputa of the distributor is shown in Figure 7. Blockage of inputa is

logically equivalent to three cases. First, if no select signal ever arrives at inputselect, the distributor
will not know how to route packets and will therefore not transmit them. Secondly, if always eventually

Figure 6: XDI state graph of the distributor. Note that several states (e.g.s11 ands12 are replicated in
order to simplify the diagram.

a 01 signal arrives on theselect wire, and if outputb is permanently blocked, eventually a packet at
input a will be permanently blocked (note that a 01 signal on theselect wire means “route towards
outputb”). The third case is similar but for outputc.

Using the lemmas presented in the previous section, theoremin Figure 7 can be proven instanta-
neously and without any interaction.

5 Conclusion

We have mechanically verified properties of a library of delay-insensitive primitives in the ACL2 theorem
prover. These properties are often deceptively simple, making it easy to formulate incorrect theorems.
Moreover, their proofs are large and cumbersome. Their formalization is tricky: it is based on XDI state

(defthm blocking-equation-distributor

(and (xdi-smp-guard *xdi-sm-distributor*)

(implies (member-equal env (reasonable-envs *xdi-sm-distributor*))

(iff (Blocked_ *xdi-sm-distributor* ’in env)

(or (and (Idle_ *xdi-sm-distributor* ’select00 env)

(Idle_ *xdi-sm-distributor* ’select01 env)

(Idle_ *xdi-sm-distributor* ’select10 env))

(and (not (Idle_ *xdi-sm-distributor* ’select01 env))

(Blocked_ *xdi-sm-distributor* ’out1 env))

(and (not (Idle_ *xdi-sm-distributor* ’select10 env))

(Blocked_ *xdi-sm-distributor* ’out0 env)))))))

Figure 7: Blocking Equation for Distributor

machines and their execution semantics relative to the environment of the primitive. Our approach con-
sists of building a checker for XDI state machines which can decide LTL formulae that are built out of
block- and idle predicates. This checker has been proven correct with respect to its specification. The the-
orems that are to be proven involve free variables and non-executable functions introduced bydefun-sk
constructs. Loading the book that contains our definitions and lemmas suffices to fully automatically
prove these theorems quickly.

The properties that have been proven are used to derive a SAT/SMT instance from an asynchronous
circuit built out of primitives in the library. This derivation can be quite contrived, especially when data
is taken into account. In the future, we plan to use ACL2 to prove correctness of our derivation, proving
that feasibility of the derived SAT/SMT instance is logically equivalent to the existence of a structural
deadlock.

References

[1] Alexander Gotmanov, Satrajit Chatterjee & Michael Kishinevsky (2011):Verifying Deadlock-Freedom of
Communication Fabrics. In Ranjit Jhala & David A. Schmidt, editors:VMCAI , Lecture Notes in Computer
Science6538, Springer, pp. 214–231. Available athttp://dx.doi.org/10.1007/978-3-642-18275-4_

16.

[2] W.C. Mallon & J.T. Udding (1998):Building finite automata from DI specifications. In: Proceedings of
the Fourth International Symposium on Advanced Research inAsynchronous Circuits and Systems., pp.
184–193, doi:10.1109/ASYNC.1998.666504.

[3] F. Ouchet, K. Morin-Allory & L. Fesquet (2010):Delay Insensitivity Does Not Mean Slope Insensitivity!
In: 16th IEEE International Symposium on Asynchronous Circuits and Systems (ASYNC’10), pp. 176–184,
doi:10.1109/ASYNC.2010.27.

[4] A. Peeters, F. te Beest, M. de Wit & W. Mallon (2010):Click elements: An implementation style for
data-driven compilation. In: Proceedings of the IEEE Symposium on Asynchronous Circuitsand Systems
(ASYNC’10), pp. 3–14, doi:10.1109/ASYNC.2010.11.

[5] Amir Pnueli (1977): The temporal logic of programs. In: 18th Annual Symposium on Foundations of
Computer Science, pp. 46–57, doi:10.1109/SFCS.1977.32.

[6] Ivan Sutherland (2012):The tyranny of the clock. Communications of the ACM55(10), pp. 35–36, doi:10.
1145/2347736.2347749.

[7] F. Verbeek & J. Schmaltz (2011):Hunting deadlocks efficiently in microarchitectural models of communica-
tion fabrics. In: Proceedings of the International Conference on Formal Methods in Computer-Aided Design
(FMCAD’11), IEEE, pp. 223–231.

[8] Freek Verbeek & Julien Schmaltz (2013):Formal Deadlock Verification for Click Circuits. In: 19th IEEE
International Symposium on Asynchronous Circuits and Systems (ASYNC’13).

[9] T. Verhoeff (1998):Analyzing specifications for delay-insensitive circuits. In: Proceedings of the Fourth In-
ternational Symposium on Advanced Research in Asynchronous Circuits and Systems, pp. 172–183, doi:10.
1109/ASYNC.1998.666503.

[10] Chao Yan, F. Ouchet, L. Fesquet & K. Morin-Allory (2011): Formal Verification of C-element Circuits.
In: 17th IEEE International Symposium on Asynchronous Circuits and Systems (ASYNC’11), pp. 55–64,
doi:10.1109/ASYNC.2011.14.

[11] Mohamed H. Zaki, Sofine Tahar & Guy Bois (2008):Formal verification of analog and mixed signal designs:
A survey. Microelectronics Journal39(12), pp. 1395–1404, doi:10.1109/IDT.2009.5404084.

http://dx.doi.org/10.1007/978-3-642-18275-4_16
http://dx.doi.org/10.1007/978-3-642-18275-4_16
http://dx.doi.org/10.1109/ASYNC.1998.666504
http://dx.doi.org/10.1109/ASYNC.2010.27
http://dx.doi.org/10.1109/ASYNC.2010.11
http://dx.doi.org/10.1109/SFCS.1977.32
http://dx.doi.org/10.1145/2347736.2347749
http://dx.doi.org/10.1145/2347736.2347749
http://dx.doi.org/10.1109/ASYNC.1998.666503
http://dx.doi.org/10.1109/ASYNC.1998.666503
http://dx.doi.org/10.1109/ASYNC.2011.14
http://dx.doi.org/10.1109/IDT.2009.5404084

	1 Introduction
	2 Formalizing Blocking and Idle Conditions
	2.1 Formalization of Click Primitives
	2.2 Execution Semantics
	2.3 Labelling States as Blocking or Idling
	2.4 Formulating Block- or Idle Conditions

	3 Model Checking Blocking and Idle Conditions
	3.1 ACL2 Overview of Automated Proof
	3.2 Implementation of Blocked and Idle
	3.3 Proof of Lemma ??

	4 Application
	5 Conclusion

